

# 汽车发动机功率校正系数的计算

王祝鹏

(交通部公路科学研究所 北京 100088)

**提要** 本文根据汽车发动机功率校正系数的有关计算公式和饱和蒸气压等计算公式,用 BASIC 语言编写了计算程序。使用该程序,可在输入原始数据后,立即得出结果,即迅速,又准确。

**关键词** 汽车 发动机 功率 校正系数 计算程序

## Computerized Calculation of Automobile Engine Power Calibration Coefficient

Wang Zhupeng

(Research Institute of Highway the Ninistr) or communications, Beijing

**Abstract** In this paper, a computer program is presented for the calculation of automobile engine power calibration coefficient referred to conventional calibrating formulae and that of saturated steam evaporation pressure. The computer programs are written in BASIC Language, which makes it possible to quickly get calculation results by inputting necessary data.

**Key Words** Automobile Engine Power Calibration coefficient  
Computer program

### 1 前言

在汽车发动机性能室内试验中,发动机的进气状态(包括进气温度、进气干空气压、水蒸气分压及进气总压)对发动机的性能有一定影响,因此同一台发动机在不同进气状态下所得的试验数据是不一致的,给性能对比带来困难。为了解决这一问题,JB3743-84“汽车发动机性能试验方法”规定了对发动机实测功率或扭矩进行校正的方法,将发动机实测功率或扭矩校正为标准进气状态下的功率或扭矩,并规定了标准进气状态值(附表)和功率校正系数计算公式。

上述试验方法颁布较早,当时我国计算机的使用尚不普及,而且校正系数的计算又非常复杂,很难用普通计算器计算,所以使用各种计算图和计算公式求得校正系数。例如,求汽油机功率校正系数时,先测得干球温度和湿球温度,然后在“抽风式湿度计用湿度图”上(JB3743-84的附录B)查出水蒸气分压 $P_w$ 值,用公式 $P_s = P - P_w$ 算出进气干空气压 $P_s$ ,然后在“汽油机功率校正系数图”上找出与进气温度 $t$ 和进气干空气压 $P_s$ 值相对应的点,将两点连成一直线,并延长与“校正系数”尺相交,其交点即为功率校正系数值。

求柴油机功率校正系数时,不仅需要算出进气干空气压  $P_s$  值,还需要事先算出柴油机特性指数  $f_m$ ,再用“柴油机功率校正系数图”找出进气温度  $t$  和进气干空气压  $P_s$  值对应点,连接两点并延长与“进气因数  $f_a$ ”尺相交,找到  $f_a$  值。另外,在“特性指数  $f_m$ ”尺上找出事先算出的  $f_m$  值对应点,连接  $f_a$  和  $f_m$  值对应点并延长与“校正系数”尺相交,其交点即为所求的柴油机功率校正系数值。

查计算图法虽简化了计算,但仍不能摆脱中间的计算和在图上找点连线等操作。近年来,我国计算机的应用很普遍,一般较复杂的运算都在微型计算机上完成,只需输入原始数据,就能很快得出所求值,简化了中间过程并提高了计算精度。采用抽风式干湿球温度计,并全部使用计算机计算发动机功率校正系数时,还需要增加水蒸气分压和湿球温度下饱和蒸汽压的计算公式。本文介绍根据这些公式而编写的汽油发动机和柴油发动机功率校正系数计算程序和程序使用说明。

## 2 汽油发动机功率校正系数计算程序

```

10 *****
20*                汽油发动机功率校正系数计算程序                *
30'*****
40 PRINT" * * 计算汽油发动机功率校正系数 * *
50 INPUT"输入干球温度 tw(°C)=" ;TW
60 INPUT"输入湿球温度 t'w(°C)=" ;TW1
70 INPUT"输入进气温度 t (°C)=" ;T
80 INPUT"输入进气总压 P (kPa)=" ;P
90 T1=273.16
100 T2=273.15+TW1
110 I1=10.79574*(1-T1/T2)
120 I2=-5.028*LOG(T2/T1)/LOG(10)
130 I3=1.50475*10(-4)*(1-10(-8.2969*(T2/T1-1)))
140 I4=0.42873*10(-3)*(10(4.76955*(1-T1/T2))-1)
150 I5=0.78614
160 I=(I1+I2+I3+I4+I5)-1
170 EW1=10I
180 PW=EW1-0.0662*(TW-TW1)
190 PS=P-PW
200 KA=(99/PS)1.2*((273+T)/298)0.6
210 PRINT"输入数据:"
220 PRINT"tw(°C)="TW
230 PRINT"t'w(°C)="TW1
240 PRINT"t (°C)="T
250 PRINT"P(kPa)="P
260 PRINT"计算数据:"

```

```

270 PRINT USING"Ew'(kPa)=####.###";EW1
280 PRINT USING"Pw(kPa)=####.###";PW
290 PRINT USING"Ps(kPa)=####.###";PS
300 PRINT USING"Ka      =####.###";KA
310 INPUT"是否打印(是:y 否:n)";X¥
320 IF X¥="n"GOTO 430
330 LPRINT"输入数据:"
340 LPRINT"tw'(°C)="TW
350 LPRINT"tw'(°C)="TW1
360 LPRINT"t (°C)="T
370 LPRINT"P (kPa)="P
380 LPRINT"计算数据:"
390 LPRINT USING"Ew'(kPa)=####.###";EW1
400 LPRINT USING"Pw(kPa)=####.###";PW
410 LPRINT USING"Ps(kPa)=####.###";PS
420 LPRINT USING"Ka =####.###";KA
430 END

```

### 3 柴油发动机功率校正系数计算程序

```

10' * * * * *
20' *                柴油发动机功率校正系数计算程序                *
30' * * * * *
40 PRINT" * * 计算柴油发动机功率校正系数 * *
50 INPUT"输入干球温度 tw(°C)=";TW
60 INPUT"输入湿球温度 tw'(°C)=";TW1
70 INPUT"输入进气温度 t (°C)=";T
80 INPUT"输入进气总压 P (kPa)=";P
90 INPUT"输入燃料消耗量 Gf(kg/h)=";GF
100 INPUT"输入发动机排量 Vh(L)=";VH
110 INPUT"输入发动机转速 n(r/min)=";N
120 INPUT"发动机冲程数(四冲程:4 二冲程:2)";C¥
130 INPUT"增压方式(非增压:f 机械增压:j 涡轮增压:w)";Z¥
140 T1=273.16
150 T2=273.15+TW1
160 I1=10.79574*(1-T1/T2)
170 I2=-5.028*LOG(T2/T1)/LOG(10)
180 I3=1.50475*10(-4)*(1-10(-8.2969*(T2/T1-1)))
190 I4=0.42873*10(-3)*(10(4.76955*(1-T1/T2))-1)
200 I5=0.78614

```

```

210 I=(I1+I2+I3+I4+I5)-1
220 EW1=10^I
230 PW=EW1-0.0662*(TW-TW1)
240 PS=P-PW
250 IF Z¥="W" GOTO 320
260 FA=99/PS*((273+T)/298)^0.7
270 IF Z¥="j" GOTO 300
280 R=1
290 GOTO 360
300 INPUT"增压器压气机进出口压比r=";R
310 GOTO 360
320 INPUT"输入涡轮增压器的压气机进口压力Pi(kPa)=";PI
330 INPUT"输入涡轮增压器的压气机出口压力Po(kPa)=";PO
340 FA=(99/PS)^0.7*((273+T)/298)^1.5
350 R=PO/PI
360 IF C¥="2" GOTO 390
370 Q=33333*GF/(VH*N)
380 GOTO 400
390 Q=16667*GF/(VH*N)
400 FM=0.036*(Q/R)-1.14
410 IF(Q/R)<40 THEN FM=0.3
420 IF(Q/R)>65 THEN FM=1.2
430 KD=FA^FM
440 PRINT"输入数据:"
450 PRINT"tw(°C)="TW
460 PRINT"tw'(°C)="TW1
470 PRINT"t(°C)="T
480 PRINT"P(kPa)="P
490 PRINT"Gf(kg/h)="GF
500 PRINT"Vh(L)="VH
510 PRINT"n(r/min)="N
520 IF Z¥="f" GOTO 580
530 IF Z¥="j" GOTO 570
540 PRINT"Pi(kPa)="PI
550 PRINT"Po(kPa)="PO
560 GOTO 580
570 PRINT"r=";R
580 PRINT"计算数据:"
590 PRINT USING"Ew'(kPa)=####.###";EW1

```

```

600 PRINT USING "Pw(kPa)=####.##";PW
610 PRINT USING "Pi(kPa)=####.##";PS
620 PRINT USING "fa =####.##";FA
630 IF Z¥="j" GOTO 650
640 PRINT USING "r=####.##";R
650 PRINT USING "q(mg/L.cyc)=####.##";Q
660 PRINT USING "fm =####.##";FM
670 PRINT USING "Kd =####.##";KD
680 INPUT "是否打印(是:y 否:n)";X¥
690 IF X¥="n" GOTO 940
700 LPRINT "输入数据:"
710 LPRINT "tw(°C)="TW
720 LPRINT "t'w(°C)="TW1
730 LPRINT "t(°C)="T
740 LPRINT "P(kPa)="P
750 LPRINT "Gf(kg/h)="GF
760 LPRINT "Vh(L)="VH
770 LPRINT "n(r/min)="N
780 IF Z¥="f" GOTO 840
790 IF Z¥="j" GOTO 830
800 LPRINT "Pi(kPa)="PI
810 LPRINT "Po(kPa)="PO
820 GOTO 840
830 LPRINT "r=";R
840 LPRINT "计算数据:"
850 LPRINT USING "Ew'(kPa)=####.##";EW1
860 LPRINT USING "Pw(kPa)=####.##";PW
870 LPRINT USING "Ps(kPa)=####.##";PS
880 LPRINT USING "fa =####.##";FA
890 IF Z¥="j" GOTO 910
900 LPRINT USING "r =####.##";R
910 LPRINT USING "q(mk/L.cyc)=####.##";Q
920 LPRINT USING "fm =####.##";FM
930 LPRINT USING "Kd =####.##";KD
940 END

```

#### 4 程序使用说明

当键入“RUN”使该程序运行后,可根据屏幕上的提示,输入原始数据。输入增压方式时,如果直接输入增压器压气机进出口压力比则选择“J”;如果输入增压器压气机进出口压力则选汽车发动机功率校正系数的计算——王祝鹏

择“W”。原始数据输入完毕,屏幕上即显示出输入数据和计算数据值。如果还需要打印,可在“是否打印?”后面选择“Y”。

根据 JB3743-84“汽车发动机性能试验方法”的要求,当  $q/r$  值小于 40 mg/L.cyc 时,程序即自动令  $f_m=0.3$ ;当  $q/r$  值大于 65 mg/L.cyc 时,程序则自动令  $f_m=1.2$ 。

以汽油发动机和非增压柴油机为例,程序运行后可以显示并打印以下数据:

①汽油发动机:

输入数据:

$t_w$  (°C) = 28  
 $t_w'$  (°C) = 22  
 $t$  (°C) = 25  
 $P$  (kPa) = 99.7

计算数据:

$E_w'$  (kPa) = 2.643  
 $P_w$  (kPa) = 2.25  
 $P_s$  (kPa) = 97.45  
 $K_a$  = 1.019

②柴油发动机:

输入数据:

$t_w$  (°C) = 28  
 $t_w'$  (°C) = 22  
 $t$  (°C) = 25  
 $P$  (kPa) = 99.7  
 $G_f$  (kg/h) = 28  
 $V_h$  (L) = 12  
 $n$  (r/min) = 1800

计算数据:

$E_w'$  (kPa) = 2.643  
 $P_w'$  (kPa) = 2.25  
 $P_s$  (kPa) = 97.45  
 $f_a$  = 1.016  
 $r$  = 1.00  
 $q$  (mg/L.cyc) = 43.21  
 $f_{in}$  = 0.42  
 $K_d$  = 1.007

5 附表及公式

5.1 汽油发动机功率校正系数计算公式:

$$M_{eo} = K_a \cdot M_e$$

$$N_{eo} = K_a \cdot N_e$$

$$K_a = \left(\frac{99}{P_s}\right)^{1.2} \cdot \left(\frac{T}{298}\right)^{0.6}$$

$$= \left(\frac{99}{P_s}\right)^{1.2} \cdot \left(\frac{273+t}{298}\right)^{0.6}$$

$$P_s = P - P_w$$

标准进气状态表 附表

进气参数	标准进气状态值
进气温度	298K
进气干空气压	99kPa
水蒸气分压	1kPa
进气总压	100kPa

式中:

$M_{eo}$ ——校正有效扭矩 (N·m);

$M_e$ ——实测有效扭矩 (N·m);

$N_{eo}$ ——校正有效功率 (kW);

$N_e$ ——实测有效功率 (kW);

$K_a$ ——汽油机校正系数;

$P_s$ ——进气干空气压 (kPa);

$T$ ——进气温度 (K);

$t$ ——进气温度 (°C);

$P$ ——进气总压 (kPa);

$P_w$ ——水蒸气分压 (kPa)。

## 5.2 柴油发动机功率校正系数计算公式:

$$M_{\infty} = K_d \cdot M_s$$

$$N_{\infty} = K_d \cdot N_s$$

$$K_d = f_s^{f_m}$$

式中:

$M_{\infty}$ ——校正有效扭矩 (N·m)

$M_s$ ——实测有效扭矩 (N·m)

$N_{\infty}$ ——校正有效功率 (kW)

$N_s$ ——实测有效功率 (kW)

$K_d$ ——柴油机校正系数

$f_s$ ——进气因数

$f_m$ ——柴油机特性指数

### 5.2.1 非增压及机械增压式进气因数:

$$f_s = \frac{99}{P_s} \cdot \left(\frac{T}{298}\right)^{0.7}$$

$$= \frac{99}{P_s} \cdot \left(\frac{273+t}{298}\right)^{0.7}$$

$$P_s = P - P_w$$

式中:

$P_s$ ——进气干空气压 (kPa)

$T$ ——进气温度 (K)

$t$ ——进气温度 (°C)

$P$ ——进气总压 (kPa)

$P_w$ ——水蒸气分压 (kPa)

### 5.2.2 涡轮增压式进气因数:

$$f_s = \left(\frac{99}{P_s}\right)^{0.7} \cdot \left(\frac{T}{298}\right)^{1.5}$$

$$= \left(\frac{99}{P_s}\right)^{0.7} \cdot \left(\frac{273+t}{298}\right)^{1.5}$$

### 5.2.3 柴油机特性指数:

$$f_m = 0.036 \frac{q}{r} - 1.14$$

式中:

$q$ ——比容循环供油量 (mg/L·cyc)

$$\text{四冲程柴油机: } q = 33333 \frac{G_f}{V_H \cdot n}$$

$$\text{二冲程柴油机: } q = 16667 \frac{G_f}{V_H \cdot n}$$

式中:

$G_f$ ——燃料消耗量 (kg/h)

$V_H$ ——发动机排量 (L)

$n$ ——发动机转速 (r/min)

$r$ ——涡轮增压器压力比。

$$\text{涡轮增压式: } r = \frac{P_0}{P_1}$$

式中:

$P_0$ ——涡轮增压器的压气机出口压力 (kPa)

$P_1$ ——涡轮增压器的压气机进口压力 (kPa)

非增压式:  $r = 1$

## 5.3 水蒸气分压和湿球温度下的

饱和蒸气压计算公式:

$$P_w = E_w' - 0.0662(t_w - t_w')$$

式中:

$P_w$ ——水蒸气分压 (kPa);

$E_w'$ ——湿球温度下的饱和蒸气压 (kPa);

$t_w$ ——干球温度 (°C);

$t_w'$ ——湿球温度 (°C)。

$$\text{Lg } E_w = \left\{ 10.79574 \left( 1 - \frac{T_1}{T_2} \right) \right.$$

$$\left. - 5.02800 \text{Lg} \left( \frac{T_2}{T_1} \right) \right.$$

$$\left. + 1.50475 \times 10^{-4} [1 - 10^{-8.2969(T_2/T_1 - 1)}] \right.$$

$$\left. + 0.42873 \times 10^{-3} [10^{4.76955(1 - T_1/T_2)} - 1] \right.$$

$$\left. + 0.78614 \right\} - 1$$

式中:

$T_1$ ——水的三相点温度 (K),

$$T_1 = 273.16$$

$T_2$ ——湿球温度 (K),

$$T_2 = 273.15 + t_w'$$